

Roikkuva ketju

Tarkastellaan kitkattoman, taipuisan ketjun potentiaalienergiaa. Jos ketju riippuu molemmista päistä kiinnitettynä, se hakee muodon, joka antaa potentiaalienergialle minimiarvon. Variaatiolaskennan avulla voidaan osoittaa, että kun ketjun potentiaalienergialla on minimi, niin ketjun muoto $y = y(x)$ noudattaa differentiaaliyhtälöä

$$\frac{d^2 y}{dx^2} = k \sqrt{1 + \left(\frac{dy}{dx}\right)^2}.$$

Tässä k on ketjun pituuden epäsuorasti määrittävä tekijä. Tarkastellaan roikkuvan ketjun muotoa eri reunaehdoilla. Poistetaan ensin vanhat muuttujat.

```
Remove["Global`*"]
```

Muodostetaan ketjua kuvaava toisen kertaluvun differentiaaliyhtälö. Valitaan $k = 1$.

```
k = 1; yht = y''[x] == k Sqrt[1 + y'[x]^2]
```

$$y''[x] = \sqrt{1 + y'[x]^2}$$

Valitaan reunaehdot symmetrisesti origon molemmin puolin. Ketju olkoon kiinnitetty pisteissä $x = 1$ ja $x = -1$.

```
reunaehto1 = {y[-1] == 0, y[1] == 0}
```

```
{y[-1] == 0, y[1] == 0}
```

Ratkaistaan yhtälö:

```
rtk1 = DSolve[Flatten[{yht, reunaehto1}], y[x], x]
```

```
Solve::ifun : Inverse functions are being used by Solve, so some
solutions may not be found; use Reduce for complete solution information. More...
```

```
{{y[x] -> Cosh[1] - Cosh[x]}, {y[x] -> -Cosh[1] + Cosh[x]}}
```


Ratkaisualgoritmissa käytetään monihaaraisia käänteisfunktioita, mistä `DSolve` antaa varoituksen. Ketjun muoto noudattaa hyperbolista kosinia, jonka kuvaajaa tästä syystä kutsutaankin *ketjukäyräksi*. Sievennetään tulosta.

```
kaari1 = y[x] /. rtk1 // FullSimplify
```

```
{Cosh[1] - Cosh[x], -Cosh[1] + Cosh[x]}
```

Valitaan ratkaisusta se, joka aukeaa ylöspäin ($\cosh(x)$) ja piirretään kuvaaja ketjun riippumisasennolle.

```
Plot[kaari1[[2]], {x, -1, 1}, AspectRatio -> Automatic]
```



- Graphics -

Entä kun ketjun toinen pää on merkittävästi korkeammalla? Muutetaan reunaehtoja ja ratkaistaan yhtälöt.

```
reunaehto2 = {y[-1] == 0, y[1] == 3}
```

```
{y[-1] == 0, y[1] == 3}
```

```
rtk2 = DSolve[Flatten[{yht, reunaehto2}], y[x], x] // N
```

Solve::ifun : Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information. [More...](#)

```
{{y[x] -> 0.5 (-2.00409 + 3.24292 Cosh[x] + 2.55275 Sinh[x])},
 {y[x] -> 0.5 (8.00409 - 3.24292 Cosh[x] + 2.55275 Sinh[x])}}
```


Jälleen epävarmuutta aiheuttava käänteisfunktio. Sievennetään tulosta:

```
kaari2 = y[x] /. rtk2 // Simplify
```

```
{-1.00205 + 1.62146 Cosh[x] + 1.27638 Sinh[x], 4.00205 - 1.62146 Cosh[x] + 1.27638 Sinh[x]}
```

Tulokseksi saadaan hieman monimutkaisempi riippumiskäyrä. Valitaan ylöspäin aukeava ratkaisu ja piirretään sen kuvaaja. (Mathematican eri versiot saattavat antaa ratkaisut eri järjestyksessä, joten on tutkittava, kumpi ratkaisu aukeaa oikeaan suuntaan.)

```
Plot[kaari2[[1]], {x, -1, 1}, AspectRatio -> Automatic]
```



- Graphics -

Kun käyrän yhtälö ilmaistaan muodossa $y = f(x)$, saadaan käyrän pituus L laskemalla

$$L = \int_{x_1}^{x_2} \sqrt{1 + y'^2} \, dx,$$

missä $y' = \frac{dy}{dx}$. Tätä yhtälöä voidaan soveltaa myös ratkaistuihin riippuvien ketjujen yhtälöihin. Lasketaan ensimmäisenä lasketun käyrän pituus. Määritellään ensin integraalin sisällä oleva funktio:

```
ketju = kaari1[[1]]; fkt[x_] = Sqrt[1 + D[ketju, x]^2]
```

$$\sqrt{1 + \sinh[x]^2}$$

Sitten integroidaan:

```
pit = NIntegrate[fkt[x], {x, -1, 1}]
```

```
2.3504
```

Tehtäviä

Portaikko suljetaan ketjulla, joka ripustetaan kahden tolpan välille. Tolppien etäisyys on 1 m ja korkeus 70 cm. Ratkaise graafisesti kokeilemalla (k :n arvoa muuttamalla), minkä pituinen ketju tarvitaan, kun ketjun alimmaiskorkeuden tulee olla 50 cm. Kun muutat k :n arvoa, saatat kohdata yllättäviä vaikeuksia differentiaaliyhtälön ratkaisussa. Johtuen Mathematican käyttämästä ratkaisurutiinista ja funktioiden haaroista kompleksitasossa, ratkaisuyhtälöt sisältävät useita päällekkäisiä juuria. Laskentaresursseista riippuen on syytä tyytyä k :n 1 – 2 desimaalin tarkkuuteen, sillä se antaa ketjun pituudelle jo alle senttimetrin tarkkuuden. Ketjun pituudeksi tulee n. 109.7 cm.

Mathematica ei osaa ratkaista riippuvan ketjun yhtälöä, mikäli k :lle ei ole määritelty lukuarvoa. Löydätkö yhteyden ketjun pituuden ja vakion k välille kokeilemalla k :lle eri arvoja?

Roikkuvan ketjun potentiaalienergiaa kuvaa integraali $\int_{x_1}^{x_2} y(x) \sqrt{1 + y'(x)^2} dx$, kun ketjun massatiheys pituusyksikköä kohden on asetettu $= 1$. Variaatiolaskennan Eulerin yhtälön mukaan tämä saa minimiarvonsa, kun $y y'' = 1 + y'^2$. Etsi tämän yhtälön yleinen ratkaisu alkeisfunktioiden avulla. Johda edellä lähtökohdaksi otettu differentiaaliyhtälö ratkaisun välitulosten avulla. Vihje: Kyseessä on toisen kertaluvun yhtälö, jossa muuttuja x ei esiinny eksplisiittisesti. Mathematicaa voi käyttää laskuissa hyödyksi, vaikka se ei osannekaan ratkaista differentiaaliyhtälöä suoraan.

Linkit

reunaehto

muotoa $y'' = f(y, y')$ olevat differentiaaliyhtälöt

JP & SKK 09.05.2001